

A Computer Program in Basic for Determining Probit and Log-Probit or Logit Correlation for Toxicology and Biology

M. M. Abou-Setta,¹ R. W. Sorrell,² and C. C. Childers^{2,*}

¹Plant Protection Research Institute, Agriculture Research Center, Giza 12611, Egypt and ²University of Florida, CREC, Lake Alfred, FL 33850

The probit function is widely used to model a logistic relationship between a stimulus and a quantal response in toxicology and biology. There are well known programs of this type for toxicology written in Fortran (Daum et al. 1966, Daum 1970, Russell et al. 1977 and Robertson et al. 1981) to run on minicomputers or main-frame computers. However, these programs are specifically intended for bio-assay and are not available for microcomputers.

The probit function may be appropriate for modeling other systems showing sigmoid relationships. The SAS (Statistical Analysis System) program has the probit function and will soon be available on IBM-PCs and compatibles, but will require a hard disk and 512 kilobytes of main memory. Many existing systems do not meet these specifications. The objective of this paper is to provide an additional means for determining two factor log-probit analysis. The BASIC programs presented here were developed using the probit equation and the tables published by Bliss (1935) and Finny (1952). These programs can be used as presented or adapted as subroutines within other application programs. There is no other available software offering the probit function for micro-computers to the best of our knowledge.

RESULTS AND DISCUSSION

The first program (PROPER) accepts values between 0.01 and 99.99 percent and transforms them to probit values or transforms probit values between 1.281 and 8.719 to percent values (Table 1). Examples of the input and output are shown in Table 2.

The second program (LOGPRO) accepts paired data and models the relationship between the pairs as

* Correspondence and reprint requests.

Table 1. Source listing of program PROPER

```

100 REM PROGRAM PROPER.BAS VERSION 1.0 RELEASED 20 MAY 1985
110 REM M. M. ABOU-SETTA, R. W. SORRELL, AND C. C. CHILDERS
250 ON ERROR GOTO 1020
260 REM CONSTANT
270 ININC = .001
280 REM PERCENT TO PROBIT TABLE
290 DIM PERCTB(17),PROBITB(17)
300 DATA .01,1.2810, .1,1.9098, 1,2.6737, 5,3.3551, 10,3.7184
310 DATA 20,4.1584, 30,4.4756, 40,4.7467, 50,5.0, 60,5.2533
320 DATA 70,5.5244, 80,5.8416, 90,6.2816, 95,6.6449, 99,7.3263
330 DATA 99.9,8.0902, 99.99,8.7190
340 FOR I=1 TO 17
350 READ PERCTB(I),PROBITB(I)
360 NEXT I
370 PRINT
380 PRINT " 1 - PERCENT TO PROBIT "
390 PRINT " 2 - PROBIT TO PERCENT "
400 INPUT " ENTER 1 OR 2 :";DT
410 IF DT=1 OR DT=2 THEN 440
420 PRINT " INVALID SELECTION"
430 GOTO 400
440 INPUT " ENTER NUMBER OF VALUES TO BE TRANSFORMED (AN INTEGER) ";N
450 IF N<>INT(N) THEN 440
460 IF N=0 THEN 1090
470 DIM PV(N)
480 PRINT " ENTER ";N;" VALUES (1 PER LINE, NO ZERO DATA) "
490 IF DT=1 THEN PRINT " BETWEEN .01 AND 99.99"
500 IF DT=2 THEN PRINT " BETWEEN 1.2810 AND 8.7190"
510 FOR C = 1 TO N
520 INPUT PV(C)
530 IF DT=1 AND (PV(C)>=.01 AND PV(C)<=99.99) THEN 570
540 IF DT=2 AND (PV(C)>=1.2810 AND PV(C)<=8.7190) THEN 570
550 PRINT " VALUE OUT OF RANGE, TRY AGAIN "
560 GOTO 520
570 NEXT C
580 IF DT=1 THEN PRINT TAB(4);"PERCENT";TAB(20);"PROBIT"
590 IF DT=2 THEN PRINT TAB(4);"PROBIT";TAB(20);"PERCENT"
600 FOR C = 1 TO N
610 REM TRANSFORMING THE PROBIT VALUE TO PERCENT
620 IF DT=1 AND PV(C)<.01 THEN 780
630 IF DT=2 AND PV(C)<1.2810 THEN 780
640 FOR I = 1 TO 16
650 IF DT=1 AND (PV(C)<PERCTB(I) OR PV(C)>=PERCTB(I+1)) THEN 700
660 IF DT=2 AND (PV(C)<PROBITB(I) OR PV(C)>=PROBITB(I+1)) THEN 700
670 SUMIN = PERCTB(I)
680 START = PROBITB(I)
690 GO TO 800
700 NEXT I
710 REM UPPER LIMITS
720 IF DT<>1 OR PV(C)<>99.99 THEN 750
730 CALCPR = 8.7190
740 GOTO 970
750 IF DT<>2 OR PV(C)<>8.719 THEN 780
760 SUMIN = 99.99
770 GOTO 980
780 PRINT PV(C);" IS OUT OF RANGE"
790 GOTO 990
800 REM INTEGRATE BY TRAPEZOIDAL RULE
810 IF DT=2 THEN 900
820 P1 = .3989422 * EXP(-(START - 5)^2 / 2) * 100
830 FOR CALCPR = START TO 10-ININC STEP ININC
840 IF SUMIN>=PV(C) THEN 970
850 P2 = .3989422 * EXP(-(CALCPR+ININC - 5)^2 / 2) * 100
860 SUMIN = SUMIN + (P1 + P2) / 2 * ININC
870 P1=P2
880 NEXT CALCPR
890 GOTO 970
900 IF START>=PV(C) THEN 980
910 P1 = .3989422 * EXP(-(START - 5)^2 / 2) * 100
920 FOR CALCPR = START TO PV(C)-ININC STEP ININC

```

Table 1 continued.

```

930 P2 = .3989422 * EXP(-(CALCPR+ININC - 5)^2 / 2) * 100
940 SUMIN = SUMIN + (P1 + P2)/ 2 * ININC
950 P1 = P2
960 NEXT CALCPR
970 IF DT=1 THEN PRINT TAB(4);PV(C);TAB(20);INT(CALCPR*10000+.5)/10000
980 IF DT=2 THEN PRINT TAB(4);PV(C);TAB(20);INT(SUMIN*1000+.5)/1000
990 NEXT C
1000 PRINT
1010 GOTO 1090
1020 REM INPUT ERROR TRAP
1030 IF ERL<>400 AND ERL<>440 AND ERL<>520 THEN 1070
1040 PRINT " INVALID ENTRY, TRY AGAIN ";
1050 RESUME
1070 PRINT " ERROR ";ERR;" OCCURRED AT LINE ";ERL
1080 RESUME 1090
1090 END

```

Table 2. Examples of input and output for PROPER.

```

1 - PERCENT TO PROBIT
2 - PROBIT TO PERCENT
ENTER 1 OR 2 :? 1
ENTER NUMBER OF VALUES TO BE TRANSFORMED (AN INTEGER) ? 5

ENTER 5 VALUES (1 PER LINE, NO ZERO DATA)
BETWEEN .01 AND 99.99
? .02
? .6
? 84.5
? 97
? 9.888

```

PERCENT	PROBIT
.02	1.46
.6	2.4888
84.5	6.0156
97	6.8809
9.888	3.7121

```

1 - PERCENT TO PROBIT
2 - PROBIT TO PERCENT
ENTER 1 OR 2 :? 2
ENTER NUMBER OF VALUES TO BE TRANSFORMED (AN INTEGER) ? 3

ENTER 3 VALUES (1 PER LINE, NO ZERO DATA)
BETWEEN 1.2810 AND 8.7190
? 1.2810
? 5.01
? 8.2135

```

PROBIT	PERCENT
1.281	.01
5.01	50.399
8.2135	99.934

log-probit (Table 3). The first value (X) is the stimulus and the second (Y) is the quantal response shown as a percent value. The output of this program can be stored in another file and contains the original input data and the transformed values (natural logarithms of the stimulus and probits of the response). The output also shows expected values so they can be used in plotting the relationship as linear (log-probit) or curvilinear (logit). The

Table 3. Source listing of program LOGPRO.

```

100 REM PROGRAM LOGPRO.BAS VERSION 1.0 RELEASED 13 MAY 1985
110 REM M. M. ABOU-SETTA, R. W. SORRELL, AND C. C. CHILDERS
420 ON ERROR GOTO 2020
430 REM DEFINE CONSTANTS
440 S$ = " "
450 ININC = .001
460 FLAG = 0
470 REM PERCENT TO PROBIT TABLE
480 DIM PERCTB(17),PROBITB(17)
490 DATA .01,1.2810, .1,1.9098, 1,2.6737, 5,3.3551, 10,3.7184
500 DATA 20,4.1584, 30,4.4756, 40,4.7467, 50,5.0, 60,5.2533
510 DATA 70,5.5244, 80,5.8416, 90,6.2816, 95,6.6449, 99,7.3263
520 DATA 99.9,8.0902, 99.99,8.7190
530 FOR I=1 TO 17
540 READ PERCTB(I),PROBITB(I)
550 NEXT I
560 PRINT
570 PRINT " LOG-PROBIT CORRELATION BETWEEN TWO FACTORS"
580 PRINT " ENTER TITLE OF NO MORE THAN 70 CHARACTERS"
590 INPUT T$
600 T$ = " " + T$
610 INPUT " ENTER NUMBER OF DATA POINTS (X,Y) ( >3 , INTEGER)";N
620 IF N<3 OR N<>INT(N) THEN 610
630 DIM PCENT(N),STIMULUS(N),LX(N),PY(N),EPROBY(N),EY(N)
640 PRINT " ENTER ";N;" PAIRS OF VALUES: STIMULUS , REACTION"
650 PRINT " STIMULUS AS A REAL NUMBER GREATER THAN 0"
660 PRINT " REACTION AS .01 TO 99.99 PERCENT"
670 FOR C = 1 TO N
680 PRINT " STIMULUS ";C;"=" " ;
690 INPUT STIMULUS(C)
700 IF STIMULUS(C)>0 THEN 730
710 PRINT " MUST BE A NUMBER GREATER THAN 0"
720 GOTO 680
730 PRINT " REACTION ";C;"=" " ;
740 INPUT PCENT(C)
750 IF PCENT(C)>=.01 AND PCENT(C)<=99.99 THEN 780
760 PRINT " MUST BE A NUMBER NOT LESS THAN .01 OR GREATER THAN 99.99"
770 GOTO 730
780 NEXT C
790 PRINT
800 PRINT " WORKING"
810 REM TRANSFORM STIMULUS TO LOG AND REACTION TO PROBIT
820 FOR C = 1 TO N
830 REM TRANSFORM THE STIMULUS TO LOG AND ACCUMULATE
840 REM SUM OF X AND SUM OF X SQUARED
850 LX(C) = LOG(STIMULUS(C))
860 SLX = SLX + LX(C)
870 SMLX2 = SMLX2 + (LX(C))^2
880 REM TRANSFORMING THE REACTION TO A PROBIT
890 IF PCENT(C)<.01 THEN 1000
900 FOR I = 1 TO 16
910 IF PCENT(C)<PERCTB(I) OR PCENT(C)>=PERCTB(I+1) THEN 950
920 ISUM = PERCTB(I)
930 BEGIN = PROBITB(I)
940 GO TO 1020
950 NEXT I
960 REM UPPER LIMITS
970 IF PCENT(C)<>99.99 THEN 1000
980 CALCPROB = 8.7190
990 GOTO 1100
1000 PRINT PCENT(C);" IS OUT OF RANGE"
1010 GOTO 2140
1020 REM INTEGRATE BY TRAPEZOIDAL RULE UNTIL PERCENT VALUE IS REACHED
1030 P1 = .3989422 * EXP(-(BEGIN - 5)^2 / 2) * 100
1040 FOR CALCPROB = BEGIN TO 10-ININC STEP ININC
1050 IF ISUM>=PCENT(C) THEN 1100
1060 P2 = .3989422 * EXP(-(CALCPROB+ININC - 5)^2 / 2) * 100
1070 ISUM = ISUM + (P1 + P2) / 2 * ININC
1080 P1 = P2
1090 NEXT CALCPROB

```

Table 3 continued.

```

1100 PY(C) = CALCPROB
1110 REM FIND SUM OF Y, SUM OF Y SQUARED, AND SUM OF XY
1120 SPY = SPY + PY(C)
1130 S2PY = S2PY + (PY(C)^2)
1140 SUMLXPY = SUMLXPY + (LX(C)*PY(C))
1150 NEXT C
1160 REM FIND R-SQUARED, Y-INTERCEPT, AND SLOPE OF REGRESSION LINE
1170 RSQUARE=(SUMLXPY-SLX*SPY/N)^2/(S2PY-SPY^2/N)*(SMLX2-SLX^2/N)
1180 NTERCEPT = (SMLX2*SPY-SUMLXPY*SLX)/(N*SMLX2-SLX^2)
1190 LSLOPE = (N*SUMLXPY-SPY*SLX)/(N*SMLX2-SLX^2)
1200 REM CALCULATING THE EXPECTED REACTION VALUES
1210 FOR C = 1 TO N
1220 EPROBY(C) = NTERCEPT+LSLOPE*(LOG(STIMULUS(C)))
1230 FOR I = 1 TO 16
1240 IF EPROBY(C)<PROBITB(I) OR EPROBY(C)>=PROBITB(I+1) THEN 1280
1250 ISUM = PERCTB(I)
1260 BEGIN = PROBITB(I)
1270 GO TO 1350
1280 NEXT I
1290 REM UPPER LIMITS
1300 IF EPROBY(C) = 8.7190 THEN ISUM = 99.99
1310 IF EPROBY(C) > 8.7190 THEN ISUM = 100
1320 IF EPROBY(C) < 1.2810 THEN ISUM = 0
1330 FLAG = 1
1340 GOTO 1430
1350 REM INTEGRATE BY TRAPEZOIDAL RULE
1360 P1 = .3989422 * EXP(-(BEGIN - 5)^2 / 2) * 100
1370 FOR CALCPROB = BEGIN TO EPROBY(C)-ININC STEP ININC
1380 IF BEGIN>=EPROBY(C) THEN 1430
1390 P2 = .3989422 * EXP(-(CALCPROB+ININC - 5)^2 / 2) * 100
1400 ISUM = ISUM + (P1 + P2) / 2 * ININC
1410 P1 = P2
1420 NEXT CALCPROB
1430 EY(C) = ISUM
1440 NEXT C
1450 PRINT
1460 REM OUTPUT TO SCREEN AND FILE
1470 L1$= " LOG-PROBIT CURVE REGRESSION AND CORRELATION"
1480 L2$= " X Y LOGe X "
1481 R2$= " PROBIT EXPECTED EXPECTED"
1490 L3$= " (STIMULUS) (REACTION%)"
1491 R3$= " Y PROBIT Y"
1500 PRINT
1510 PRINT L1$
1520 PRINT
1530 PRINT T$
1540 PRINT
1550 PRINT L2$;R2$
1560 PRINT L3$;R3$
1570 PRINT
1580 FOR C = 1 TO N
1590 PRINT TAB(2);STIMULUS(C);TAB(13);PCENT(C);
1600 PRINT TAB(24);INT(LX(C)*10000+.5)/10000;
1610 PRINT TAB(35);INT(PY(C)*10000+.5)/10000;
1620 PRINT TAB(46);INT(EPROBY(C)*10000+.5)/10000;
1630 PRINT TAB(57);INT(EY(C)*1000+.5)/1000
1640 NEXT C
1650 PRINT
1660 PRINT " PROB Y = ";NTERCEPT;" + ";LSLOPE;" LOGe X"
1670 PRINT
1680 PRINT " R-SQUARED =" ;RSQUARE
1690 PRINT
1700 IF FLAG=0 THEN 1730
1710 PRINT " EXPECTED Y OF 0 OR 100 IS AN ESTIMATION"
1720 PRINT " DUE TO EXTREME EXPECTED PROBIT VALUES."
1730 PRINT
1740 PRINT " IF OUTPUT IS TO BE SAVED IN A FILE, ENTER FILENAME"
1750 INPUT " IF NOT, PRESS RETURN ";FILE$
1760 IF FILE$="" THEN 2140

```

Table 3 continued.

```

1770 OPEN FILE$ FOR OUTPUT AS #1
1780 PRINT #1, S$
1790 PRINT #1, L1$
1800 PRINT #1, S$
1810 PRINT #1, T$
1820 PRINT #1, S$
1830 PRINT #1, L2$;R2$
1840 PRINT #1, L3$;R3$
1850 PRINT #1, S$
1860 FOR C = 1 TO N
1870 PRINT #1, TAB(2);STIMULUS(C);TAB(13);PCENT(C);
1880 PRINT #1, TAB(24);INT(LX(C)*10000+.5)/10000;
1890 PRINT #1, TAB(35);INT(PY(C)*10000+.5)/10000;
1900 PRINT #1, TAB(46);INT(EPROBY(C)*10000+.5)/10000;
1910 PRINT #1, TAB(57);INT(EY(C)*1000+.5)/1000
1920 NEXT C
1930 PRINT #1, S$
1940 PRINT #1, " PROB Y = ";NTERCEPT;" + ";LSLOPE;" LOGe X"
1950 PRINT #1, S$
1960 PRINT #1, " R-SQUARED =";RSQUARE
1970 PRINT #1, S$
1980 IF FLAG=0 THEN 2140
1990 PRINT #1, " EXPECTED Y OF 0 OR 100 IS AN ESTIMATION"
2000 PRINT #1, " DUE TO EXTREME EXPECTED PROBIT VALUES."
2010 GOTO 2140
2020 REM ERROR TRAPS
2030 REM FILE OPEN ERRORS
2040 IF ERL <> 1770 THEN 2070
2050 PRINT " ERROR OPENING FILE ";FILES$
2060 RESUME 1740
2070 REM INPUT ERRORS
2080 IF ERL=590 OR ERL=610 OR ERL=690 OR ERL=740 OR ERL=1750 THEN 2120
2090 REM OTHER ERRORS
2100 PRINT " ERROR ";ERR;" OCCURED AT LINE ";ERL
2110 RESUME 2140
2120 PRINT " INVALID ENTRY, TRY AGAIN ";
2130 RESUME
2140 END

```

output contains the determination factor (R-squared) of the data. An example of the input and output of the program LOGPRO is shown in Table 4.

The two programs were written in "generic" BASIC for portability. More efficient code could increase speed and decrease program size for a specific computer. The programs will run with no modification on DEC VAX minicomputers in VAX-11 BASIC, on IBM-PC microcomputers and compatibles and on the Radio Shack Model 100 with 32K of main memory. The output of LOGPRO will be difficult to read on the 40 column screen of the Model 100 and should be saved in a file and printed.

The accuracy of these programs is to plus or minus one integration step (.001) for probits. More accuracy could be obtained by decreasing the step size but this would also increase running time.

Acknowledgements. In partial fulfillment of requirements for Ph.D. (MMAS). Florida Agriculture Experiment Station Journal Series No. 6715.

Table 4. Example of input and output for LOGPRO.

LOG-PROBIT CORRELATION BETWEEN TWO FACTORS
 ENTER TITLE OF NO MORE THAN 70 CHARACTERS
 ? Sample data for LOGPRO
 ENTER NUMBER OF DATA POINTS (X,Y) (>3 , INTEGER)? 10

ENTER 10 PAIRS OF VALUES: STIMULUS , REACTION
 STIMULUS AS A REAL NUMBER GREATER THAN 0
 REACTION AS .01 TO 99.99 PERCENT
 STIMULUS 1 = ? 20
 REACTION 1 = ? .02
 STIMULUS 2 = ? 40
 REACTION 2 = ? .03
 STIMULUS 3 = ? 60
 REACTION 3 = ? 5.1
 STIMULUS 4 = ? 80
 REACTION 4 = ? 14.3
 STIMULUS 5 = ? 100
 REACTION 5 = ? 46.7
 STIMULUS 6 = ? 120
 REACTION 6 = ? 47.1
 STIMULUS 7 = ? 140
 REACTION 7 = ? 82
 STIMULUS 8 = ? 160
 REACTION 8 = ? 87
 STIMULUS 9 = ? 180
 REACTION 9 = ? 85.5
 STIMULUS 10 = ? 200
 REACTION 10 = ? 89
 WORKING

LOG-PROBIT CURVE REGRESSION AND CORRELATION

Sample data for LOGPRO

X (STIMULUS)	Y (REACTION%)	LOGe X	PROBIT Y	EXPECTED PROBIT	EXPECTED Y
20	.02	2.9957	1.46	.8061	0
40	.03	3.6889	1.569	2.4783	.583
60	5.1	4.0943	3.3651	3.4565	6.131
80	14.3	4.382	3.9334	4.1505	19.778
100	46.7	4.6052	4.9177	4.6889	37.775
120	47.1	4.7875	4.9277	5.1287	55.093
140	82	4.9416	5.9156	5.5006	69.159
160	87	5.0752	6.1266	5.8227	79.458
180	85.5	5.193	6.0586	6.1069	86.577
200	89	5.2983	6.2266	6.3611	91.317

PROB Y = -6.42113 + 2.4125 LOGe X

R-SQUARED = .943429

EXPECTED Y OF 0 OR 100 IS AN ESTIMATION
 DUE TO EXTREME EXPECTED PROBIT VALUES.

IF OUTPUT IS TO BE SAVED IN A FILE, ENTER FILENAME
 IF NOT, PRESS RETURN ? sample.lis

REFERENCES

Bliss CI (1935) The calculation of the dosage-mortality curve. Ann Appl Biol 22:134-167
 Daum RJ, Killcreas W (1966) Two computer programs for probit analysis. Bull Entomol Soc Am 12(4):365-369
 Daum RJ (1970) Revision of two computer programs for

probit analysis. Bull Entomol Soc Am 16(1):10-15
Finny DJ (1952) Probit analysis : A statistical
treatment of the sigmoid curve. Cambridge University
Press. pp 318
Robertson JL, Russell RM, Savin NE (1981) Polo2 : A
computer program for multiple probit or logit
analysis. Bull Entomol Soc Am 27(3):210-211
Russell RM, Robertson RM, Savin NE (1977) Polo : A new
computer program for probit analysis. Bull Entomol
Soc Am 23(3):209-213
Received June 21, 1985; Accepted July 15, 1985